

## E01RAF – NAG Fortran Library Routine Document

**Note.** Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

### 1 Purpose

E01RAF produces, from a set of function values and corresponding abscissae, the coefficients of an interpolating rational function expressed in continued fraction form.

### 2 Specification

```
SUBROUTINE E01RAF(N, X, F, M, A, U, IW, IFAIL)
  INTEGER          N, M, IW(N), IFAIL
  real           X(N), F(N), A(N), U(N)
```

### 3 Description

E01RAF produces the parameters of a rational function  $R(x)$  which assumes prescribed values  $f_i$  at prescribed values  $x_i$  of the independent variable  $x$ , for  $i = 1, 2, \dots, n$ . More specifically, E01RAF determines the parameters  $a_j$ , for  $j = 1, 2, \dots, m$  and  $u_j$ ,  $j = 1, 2, \dots, m - 1$ , in the continued fraction

$$R(x) = a_1 + R_m(x) \tag{1}$$

where

$$R_i(x) = \frac{a_{m-i+2}(x - u_{m-i+1})}{1 + R_{i-1}(x)}, \quad \text{for } i = m, m - 1, \dots, 2,$$

and

$$R_1(x) = 0,$$

such that  $R(x_i) = f_i$ , for  $i = 1, 2, \dots, n$ . The value of  $m$  in (1) is determined by the routine; normally  $m = n$ . The values of  $u_j$  form a re-ordered subset of the values of  $x_i$  and their ordering is designed to ensure that a representation of the form (1) is determined whenever one exists.

The subsequent evaluation of (1) for given values of  $x$  can be carried out using E01RBF.

The computational method employed in E01RAF is the modification of the Thacher–Tukey algorithm described in Graves-Morris and Hopkins [1].

### 4 References

- [1] Graves-Morris P R and Hopkins T R (1981) Reliable rational interpolation *Numer. Math.* **36** 111–128

### 5 Parameters

- 1:** N — INTEGER *Input*  
*On entry:*  $n$ , the number of data points.  
*Constraint:*  $N > 0$ .
- 2:** X(N) — *real* array *Input*  
*On entry:* X( $i$ ) must be set to the value of the  $i$ th data abscissa,  $x_i$ , for  $i = 1, 2, \dots, n$ .  
*Constraint:* the X( $i$ ) must be distinct.
- 3:** F(N) — *real* array *Input*  
*On entry:* F( $i$ ) must be set to the value of the data ordinate,  $f_i$ , corresponding to  $x_i$ , for  $i = 1, 2, \dots, n$ .

- 4:** M — INTEGER *Output*  
*On exit:*  $m$ , the number of terms in the continued fraction representation of  $R(x)$ .
- 5:** A(N) — *real* array *Output*  
*On exit:* the value of the parameter  $a_j$  in  $R(x)$ , for  $j = 1, 2, \dots, m$ . The remaining elements of A, if any, are set to zero.
- 6:** U(N) — *real* array *Output*  
*On exit:* the value of the parameter  $u_j$  in  $R(x)$ , for  $j = 1, 2, \dots, m-1$ . The  $u_j$  are a permuted subset of the elements of X. The remaining  $n - m + 1$  locations contain a permutation of the remaining  $x_i$ , which can be ignored.
- 7:** IW(N) — INTEGER array *Workspace*
- 8:** IFAIL — INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.  
*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL = 1

On entry,  $N \leq 0$ .

IFAIL = 2

At least one pair of the values  $X(i)$  are equal (or so nearly so that a subsequent division will inevitably cause overflow).

IFAIL = 3

A continued fraction of the required form does not exist.

## 7 Accuracy

Usually, it is not the accuracy of the coefficients produced by this routine which is of prime interest, but rather the accuracy of the value of  $R(x)$  that is produced by the associated routine E01RBF when subsequently it evaluates the continued fraction (1) for a given value of  $x$ . This final accuracy will depend mainly on the nature of the interpolation being performed. If interpolation of a ‘well-behaved smooth’ function is attempted (and provided the data adequately represents the function), high accuracy will normally ensue, but, if the function is not so ‘smooth’ or extrapolation is being attempted, high accuracy is much less likely. Indeed, in extreme cases, results can be highly inaccurate.

There is no built-in test of accuracy but several courses are open to the user to prevent the production or the acceptance of inaccurate results.

- (1) If the origin of a variable is well outside the range of its data values, the origin should be shifted to correct this; and, if the new data values are still excessively large or small, scaling to make the largest value of the order of unity is recommended. Thus, normalisation to the range  $-1.0$  to  $+1.0$  is ideal. This applies particularly to the independent variable; for the dependent variable, the removal of leading figures which are common to all the data values will usually suffice.
- (2) To check the effect of rounding errors engendered in the routines themselves, E01RAF should be re-entered with  $x_1$  interchanged with  $x_i$  and  $f_1$  with  $f_i$ , ( $i \neq 1$ ). This will produce a completely different vector  $a$  and a re-ordered vector  $u$ , but any change in the value of  $R(x)$  subsequently produced by E01RBF will be due solely to rounding error.

- (3) Even if the data consist of calculated values of a formal mathematical function, it is only in exceptional circumstances that bounds for the interpolation error (the difference between the true value of the function underlying the data and the value which would be produced by the two routines if exact arithmetic were used) can be derived that are sufficiently precise to be of practical use. Consequently, the user is recommended to rely on comparison checks: if extra data points are available, the calculation may be repeated with one or more data pairs added or exchanged, or alternatively, one of the original data pairs may be omitted. If the algorithms are being used for extrapolation, the calculations should be performed repeatedly with the 2,3,... nearest points until, hopefully, successive values of  $R(x)$  for the given  $x$  agree to the required accuracy.

## 8 Further Comments

The time taken by the routine is approximately proportional to  $n^2$ .

The continued fraction (1) when expanded produces a rational function in  $x$ , the degree of whose numerator is either equal to or exceeds by unity that of the denominator. Only if this rather special form of interpolatory rational function is needed explicitly, would this routine be used without subsequent entry (or entries) to E01RBF.

## 9 Example

This example program reads in the abscissae and ordinates of 5 data points and prints the parameters  $a_j$  and  $u_j$  of a rational function which interpolates them.

### 9.1 Program Text

**Note.** The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      E01RAF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          N
      PARAMETER       (N=5)
      INTEGER          NIN, NOUT
      PARAMETER       (NIN=5,NOUT=6)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, M
*      .. Local Arrays ..
      real            A(N), F(N), U(N), X(N)
      INTEGER          IW(N)
*      .. External Subroutines ..
      EXTERNAL        E01RAF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'E01RAF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) (X(I),I=1,N)
      READ (NIN,*) (F(I),I=1,N)
      IFAIL = 0
*
      CALL E01RAF(N,X,F,M,A,U,IW,IFAIL)
*
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'The values of U(J) are'
      WRITE (NOUT,99999) (U(I),I=1,M-1)
      WRITE (NOUT,*)

```

```
      WRITE (NOUT,*) 'The Thiele coefficients A(J) are'  
      WRITE (NOUT,99999) (A(I),I=1,M)  
      STOP  
*  
99999 FORMAT (1X,1P,4e12.4,/)   
      END
```

## 9.2 Program Data

```
E01RAF Example Program Data  
  0.0    1.0    2.0    3.0    4.0  
  4.0    2.0    4.0    7.0   10.4
```

## 9.3 Program Results

E01RAF Example Program Results

The values of U(J) are

```
  0.0000E+00  3.0000E+00  1.0000E+00
```

The Thiele coefficients A(J) are

```
  4.0000E+00  1.0000E+00  7.5000E-01 -1.0000E+00
```

---